Affine Connection and Curvature

The Schwarzschild Metric

Initialization

In[ ]:=
```
Needs["xAct`xCoba`"]
```

```
------------------------------------------------------------

Package xAct`xPerm`  version 1.2.3, {2015, 8, 23}

CopyRight (C) 2003-2018, Jose M. Martin-Garcia, under the General Public License.

Connecting to external linux executable...

Connection established.

------------------------------------------------------------

Package xAct`xTensor`  version 1.1.3, {2018, 2, 28}

CopyRight (C) 2002-2018, Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

Package xAct`xCoba`  version 0.8.4, {2018, 2, 28}

CopyRight (C) 2005-2018, David Yllanes and
  Jose M. Martin-Garcia, under the General Public License.

------------------------------------------------------------

These packages come with ABSOLUTELY NO WARRANTY; for details type
  Disclaimer[]. This is free software, and you are welcome to redistribute
  it under certain conditions. See the General Public License for details.

------------------------------------------------------------
```

In[ ]:=
```
$PrePrint = ScreenDollarIndices;
$DefInfoQ = False;
$UndefInfoQ = False;
```

In[ ]:=
```
DefManifold[M, 4, {λ, μ, ν, ρ, σ, α, β, γ, δ}];
dimM  = DimOfManifold[M];
dimM1 = dimM - 1;
```

Here is the definition of the coordinate system, and the metric:

Simply define the list coords = {....} and the matrix gmatrix

```
In[ ]:= coords = {t[], r[], θ[], ϕ[]};
DefConstantSymbol[mass, PrintAs → "M"];
(*DefScalarFunction[ascale,PrintAs→"a"];
Use as e.g. ascale[t[],r[]] for a function of (t,r) *)
gmatrix = DiagonalMatrix[
   {-1 + 2 mass/r[], 1/(1 - 2 mass/r[]), r[]^2, r[]^2 Sin[θ[]]^2}
  ];
DefChart[ch, M, {0, 1, 2, 3}, coords, ChartColor → Blue];
g = CTensor[gmatrix, {-ch, -ch}];
SetCMetric[g, ch, SignatureOfMetric → {3, 1, 0}];
CD = CovDOfMetric[g];
```

```
In[ ]:= Print[
  "gμν= ", ComponentArray[g[{-μ, -ch}, {-ν, -ch}]] // MatrixForm, "     ,         ",
  "g^μν= ", ComponentArray[g[{ μ,  ch}, { ν,  ch}]] // MatrixForm, "\n",
  "g = ", Determinant[g, ch][], " = ",
  Det[ComponentArray[g[{-μ, -ch}, {-ν, -ch}]]] // Simplify
 ]
```

$$g_{\mu\nu}= \begin{pmatrix} -1+\frac{2M}{r} & 0 & 0 & 0 \\ 0 & \frac{1}{1-\frac{2M}{r}} & 0 & 0 \\ 0 & 0 & r^2 & 0 \\ 0 & 0 & 0 & r^2 \operatorname{Sin}[\theta]^2 \end{pmatrix} \quad , \quad g^{\mu\nu}= \begin{pmatrix} \frac{r}{2M-r} & 0 & 0 & 0 \\ 0 & 1-\frac{2M}{r} & 0 & 0 \\ 0 & 0 & \frac{1}{r^2} & 0 \\ 0 & 0 & 0 & \frac{\operatorname{Csc}[\theta]^2}{r^2} \end{pmatrix}$$

$$g = -r^4 \operatorname{Sin}[\theta]^2 = -r^4 \operatorname{Sin}[\theta]^2$$

Affine Connection

Print nonzero components:

$\Gamma^{\lambda}{}_{\mu\nu} = \frac{1}{2} g^{\lambda\sigma}(\partial_\mu g_{\sigma\nu} + \partial_\nu g_{\sigma\mu} - \partial_\sigma g_{\mu\nu})$

The components of the Christoffel symbols are collected by the ComponentArray[expr] function.

In[ ]:=

```
list = ComponentArray[Christoffel[CD, PDch][{α, ch}, {-β, -ch}, {-γ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k]], 0],
    {Subscript[Superscript["Γ", i - 1], j - 1, k - 1], list[[i, j, k]]}
   ],
   {i, 1, Length[list]}, {j, 1, Length[list]}, {k, 1, j}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$\Gamma^{0}{}_{1,0}$	$-\frac{M}{2 M r - r^2}$
$\Gamma^{1}{}_{0,0}$	$\frac{M(-2 M + r)}{r^3}$
$\Gamma^{1}{}_{1,1}$	$\frac{M}{2 M r - r^2}$
$\Gamma^{1}{}_{2,2}$	$2 M - r$
$\Gamma^{1}{}_{3,3}$	$(2 M - r) \mathrm{Sin}[\theta]^2$
$\Gamma^{2}{}_{2,1}$	$\frac{1}{r}$
$\Gamma^{2}{}_{3,3}$	$-\mathrm{Cos}[\theta]\, \mathrm{Sin}[\theta]$
$\Gamma^{3}{}_{3,1}$	$\frac{1}{r}$
$\Gamma^{3}{}_{3,2}$	$\mathrm{Cot}[\theta]$

Curvature

Print nonzero components of Riemann: $R^{\mu}{}_{\nu\rho\sigma}$

$R^{\lambda}{}_{\rho\mu\nu} = \partial_{\mu} \Gamma^{\lambda}{}_{\nu\rho} - \partial_{\nu} \Gamma^{\lambda}{}_{\mu\rho} + \Gamma^{\lambda}{}_{\mu\sigma} \Gamma^{\sigma}{}_{\nu\rho} - \Gamma^{\lambda}{}_{\nu\sigma} \Gamma^{\sigma}{}_{\mu\rho}$ (Carroll+Hartle’s convention)

xCoba has Wald’s convention, which for a Levi-Civita Connection gives the same reult after some index raising/lowering.

The components of the Riemann tensor are collected by the ComponentArray[expr] function.

In[ ]:=
```
list = ComponentArray[Riemann[CD][{α, ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0], {Subscript[Superscript["R", i - 1], j - 1, k - 1, l - 1]
     , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, Length[list]}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R^{0}{}_{1,1,0}$	$\frac{2 M}{(2 M-r) r^2}$
$R^{0}{}_{2,2,0}$	$\frac{M}{r}$
$R^{0}{}_{3,3,0}$	$\frac{M \mathrm{Sin}[\theta]^2}{r}$
$R^{1}{}_{0,1,0}$	$\frac{2 M (2 M-r)}{r^4}$
$R^{1}{}_{2,2,1}$	$\frac{M}{r}$
$R^{1}{}_{3,3,1}$	$\frac{M \mathrm{Sin}[\theta]^2}{r}$
$R^{2}{}_{0,2,0}$	$\frac{M (-2 M+r)}{r^4}$
$R^{2}{}_{1,2,1}$	$\frac{M}{(2 M-r) r^2}$
$R^{2}{}_{3,3,2}$	$-\frac{2 M \mathrm{Sin}[\theta]^2}{r}$
$R^{3}{}_{0,3,0}$	$\frac{M (-2 M+r)}{r^4}$
$R^{3}{}_{1,3,1}$	$\frac{M}{(2 M-r) r^2}$
$R^{3}{}_{2,3,2}$	$\frac{2 M}{r}$

The Riemann tensor with all lower indices: $R_{\mu\nu\rho\sigma}$

In[ ]:=

```
list = ComponentArray[Riemann[CD][{-α, -ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list⟦i, j, k, l⟧, 0], {Subscript["R", i - 1, j - 1, k - 1, l - 1]
      , list⟦i, j, k, l⟧}
   ],
   {i, 1, Length[list]}, {j, 1, i - 1}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R_{1,0,1,0}$	$-\frac{2 M}{r^3}$
$R_{2,0,2,0}$	$\frac{M(-2 M+r)}{r^2}$
$R_{2,1,2,1}$	$\frac{M}{2 M-r}$
$R_{3,0,3,0}$	$-\frac{M(2 M-r) \mathrm{Sin}[\theta]^2}{r^2}$
$R_{3,1,3,1}$	$\frac{M \mathrm{Sin}[\theta]^2}{2 M-r}$
$R_{3,2,3,2}$	$2 M r \mathrm{Sin}[\theta]^2$

The Riemann tensor with all upper indices: $R^{\mu\nu\rho\sigma}$

In[ ]:=
```
list = ComponentArray[Riemann[CD][{α, ch}, {β, ch}, {γ, ch}, {δ, ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0],
    {Superscript[Superscript[Superscript[Superscript["R", i - 1], j - 1], k - 1], l - 1]
      , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, i - 1}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$R^{1^{0^{1^{0}}}}$	$-\frac{2\,M}{r^3}$
$R^{2^{0^{2^{0}}}}$	$-\frac{M}{(2\,M-r)\,r^4}$
$R^{2^{1^{2^{1}}}}$	$\frac{M\,(2\,M-r)}{r^6}$
$R^{3^{0^{3^{0}}}}$	$-\frac{M\,Csc[\theta]^2}{(2\,M-r)\,r^4}$
$R^{3^{1^{3^{1}}}}$	$\frac{M\,Csc[\theta]^2\,(2\,M-r)}{r^6}$
$R^{3^{2^{3^{2}}}}$	$\frac{2\,M\,Csc[\theta]^2}{r^7}$

The Ricci tensor: $R_{\mu\nu}$

In[ ]:=
```
list = ComponentArray[Ricci[CD][{-α, -ch}, {-β, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j]], 0], {Subscript["R", i - 1, j - 1]
      , list[[i, j]]}
   ],
   {i, 1, Length[list]}, {j, 1, i}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

```
{}
```

Ricci scalar:

In[ ]:=
```
Print["R = ", RicciScalar[CD][]]
```

```
R = 0
```

R^2 scalar

In[ ]:=
```
Print["R² = ", Kretschmann[CD][]]
```

$R^2 = \frac{48\,M^2}{r^6}$

Einstein tensor: $G_{\mu\nu}$

In[ ]:=
```
list = ComponentArray[Einstein[CD][{-α, -ch}, {-β, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j]], 0], {Subscript["G", i - 1, j - 1]
      , list[[i, j]]}
   ],
   {i, 1, Length[list]}, {j, 1, i}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

{}

Weyl tensor:

In[ ]:=
```
list = ComponentArray[Weyl[CD][{-α, -ch}, {-β, -ch}, {-γ, -ch}, {-δ, -ch}]];
list = Table[
   If[
    UnsameQ[list[[i, j, k, l]], 0], {Subscript["C", i - 1, j - 1, k - 1, l - 1]
      , list[[i, j, k, l]]}
   ],
   {i, 1, Length[list]}, {j, 1, i - 1}, {k, 1, Length[list]}, {l, 1, k - 1}
  ];
Partition[DeleteCases[Flatten[list], Null], 2] // TableForm
```

Out[ ]//TableForm=

$C_{1,0,1,0}$	$-\frac{2\,M}{r^3}$
$C_{2,0,2,0}$	$\frac{M\,(-2\,M+r)}{r^2}$
$C_{2,1,2,1}$	$\frac{M}{2\,M-r}$
$C_{3,0,3,0}$	$-\frac{M\,(2\,M-r)\,\mathrm{Sin}[\theta]^2}{r^2}$
$C_{3,1,3,1}$	$\frac{M\,\mathrm{Sin}[\theta]^2}{2\,M-r}$
$C_{3,2,3,2}$	$2\,M\,r\,\mathrm{Sin}[\theta]^2$

Geodesic Equations

In[ ]:=
```
DefTensor[u[μ], M];
```

The geodesic equations are:

$$\frac{du^{\mu}}{d\tau} + \Gamma^{\mu}{}_{\nu\rho}\, u^{\nu}\, u^{\rho} = 0$$

geqs is a list with the second term geqs[μ+1]=$\Gamma^{\mu}{}_{\nu\rho}\, u^{\nu}\, u^{\rho}$ for each μ=0,..,d-1.

(careful, in the expressions below, when e.g. $u^{3^2} = (u^3)^2$, the squared terms don't appear nicely)

In[ ]:=
```
geqs = ComponentArray[Christoffel[CD, PDch][{μ, ch}, {-ν, -ch}, {-ρ, -ch}]
        u[{ν, ch}] u[{ρ, ch}]] // ContractBasis // Simplify;
For[i = 1, i ≤ Length[geqs], i++, Print["d/dτ", u[{i - 1, ch}], "+(", geqs[[i]], ")=0"]]
```

$$\frac{d}{d\tau}u^{0}+\left(-\frac{2\,M\,u^{0}\,u^{1}}{2\,M\,r-r^{2}}\right)=0$$

$$\frac{d}{d\tau}u^{1}+\left(\frac{-M\,(-2\,M+r)^{2}\,u^{0^{2}}+r^{2}\left(M\,u^{1^{2}}+r\,(-2\,M+r)^{2}\left(u^{2^{2}}+\mathrm{Sin}[\theta]^{2}\,u^{3^{2}}\right)\right)}{(2\,M-r)\,r^{3}}\right)=0$$

$$\frac{d}{d\tau}u^{2}+\left(\frac{2\,u^{1}\,u^{2}}{r}-\mathrm{Cos}[\theta]\,\mathrm{Sin}[\theta]\,u^{3^{2}}\right)=0$$

$$\frac{d}{d\tau}u^{3}+\left(\frac{2\left(u^{1}+\mathrm{Cot}[\theta]\,r\,u^{2}\right)u^{3}}{r}\right)=0$$

In[ ]:=
```
DefScalarFunction[dt  , PrintAs → "ṫ"];
DefScalarFunction[dr  , PrintAs → "ṙ"];
DefScalarFunction[dθ  , PrintAs → "θ̇"];
DefScalarFunction[dϕ  , PrintAs → "ϕ̇"];
DefScalarFunction[ddt, PrintAs → "ẗ"];
DefScalarFunction[ddr, PrintAs → "r̈"];
DefScalarFunction[ddθ, PrintAs → "θ̈"];
DefScalarFunction[ddϕ, PrintAs → "ϕ̈"];
u  = CTensor[{dt[] , dr[] , dθ[]  , dϕ[] }, {ch}];
du = CTensor[{ddt[], ddr[], ddθ[], ddϕ[]}, {ch}];
{u, du}
```

Out[ ]=
```
{CTensor[{ṫ[], ṙ[], θ̇[], ϕ̇[]}, {ch}, 0], CTensor[{ẗ[], r̈[], θ̈[], ϕ̈[]}, {ch}, 0]}
```

```
In[ ]:= geqs = ComponentArray[Christoffel[CD, PDch][{μ, ch}, {-ν, -ch}, {-ρ, -ch}]
            u[{ν, ch}] u[{ρ, ch}]] // ContractBasis // Simplify;
        For[i = 1, i ≤ Length[geqs], i++, Print[du[{i - 1, ch}], "+(", geqs[[i]], ")=0"]]
```

$$\ddot{t}[]+\left(-\frac{2\,M\,\dot{r}[]\,\dot{t}[]}{2\,M\,r-r^2}\right)=0$$

$$\ddot{r}[]+\left(-\frac{2\,M^2\,\dot{t}[]^2}{r^3}-\dot{\theta}[]^2\,r+M\left(2\,\dot{\theta}[]^2+\frac{\dot{t}[]^2}{r^2}+\frac{\dot{r}[]^2}{2\,M\,r-r^2}\right)+\dot{\phi}[]^2\,(2\,M-r)\,\mathrm{Sin}[\theta]^2\right)=0$$

$$\ddot{\theta}[]+\left(\frac{2\,\dot{r}[]\,\dot{\theta}[]}{r}-\mathrm{Cos}[\theta]\,\dot{\phi}[]^2\,\mathrm{Sin}[\theta]\right)=0$$

$$\ddot{\phi}[]+\left(\frac{2\,\dot{\phi}[]\left(\dot{r}[]+\mathrm{Cot}[\theta]\,\dot{\theta}[]\,r\right)}{r}\right)=0$$

Acknowledgements

This notebook has been programmed by Konstantinos Anagnostopoulos, Physics Department, National Technical University of Athens, Greece, while he was an instructor of the 4th year undergraduate course "General Relativity and Cosmology". It was created for fun, but it may turn out to be useful to everyone studying the General Theory of Relativity for the first time.

Email: konstant@mail.ntua.gr

Web: http://physics.ntua.gr/konstant